POGLAVJE II

II.3 Reševanje diferencialne enačbe lastnega nihanja

1. Reševanje diferencialnih enačb s programom Mathematica
2. Reševanje diferencialne enačbe lastnega nedušenega nihanja
3. Reševanje diferencialne enačbe lastnega dušenega nihanja
4. Prikaz razlike med odzivoma nedušenega in dušenega lastnega nihanja

1. Reševanje diferencialnih enačb s programom *Mathematica*

Tudi navadne diferencialne enačbe se programskim paketom *Mathematica* rešujejo z ukazom *DSolve[diferencialna enačba, funkcija, neodvisni parameter]*. *Mathematica* poskusi poiskati rešitev v analitični obliki, zapiše pa tudi neznane (integracijske) konstante v obliki C[1], C[2],.... (ki jih pri nedoločenem integralu ne zapiše).
Mogoče pa je, hkrati z diferencialno enačbo, podati tudi potrebne informacije (robne pogoje), ki omogočajo zapis rešitve brez neznanih konstant.
Reševanje diferencialnih enačb bo prikazano na dveh primerih, za iskanje nedušenega in dušenega odziva linearnega sistema.

2. Reševanje diferencialne enačbe lastnega nedušenega nihanja

Ker se bosta rešitvi nedušenega in dušenega nihanja poiskali v isti datoteki (zaradi njune primerjave), se zato v novo datoteko (notebook) npr. najprej zapiše v prvo vrstico naslov sekcije Nedušeni odziv, nato pa se z ukazom *Format/Style/Section* spremeni iz navadne vhodne vrstice v naslov sekcije. Vse celice te sekcije so nato objete z dodatno vertikalno črto na desnem robu, ki nakazuje pripadnost celici:



Sekcijo označuje kvadrat pred imenom:

■ Nedušeni odziv

Diferencialna enačba lastnega nedušenega nihanja se zapiše kot (analogija prvi navadni diferencialni enačbi, ki nastopi pri reševanju parcialne diferencialne enačbe, podpoglavje II.1):

$m \cdot \ddot{u} + k \cdot u = 0$

kjer sta *m* in *k* (posplošena) masa in (posplošena) togost sistema, u≡u(t) pa iskani pomik, ki je seveda funkcija časa. Ker gre za diferencialno enačbo drugega reda, sta potrebni tudi dve informaciji (robna pogoja), ki omogočata določitev neznanih integracijskih konstant. Najbolj pogosto gre za začetni informaciji oz. začetna pogoja, ki sta v obravnavani enačbi največ prva odvoda neznane funkcije. Najbolj pogosta splošna pogoja zajameta začetni (ob času t=0) pomik $u(0) \equiv u_o$ in pripadajočo začetno hitrost $\dot{u}(0) = v_o$. Ta dva pogoja je mogoče v *Mathematici* kar zapisati hkrati z diferencialno enačbo med zavita oklepaja v obliki *{diferencialna enačba, robni pogoji}*:

```
In[1]:= DSolve[{m  u''[t] + k u[t] == 0, u[0] == u0, u'[0] == v0}, u[t], t]
```

Out[1]= $\left\{\left\{u[t] \to u0\ \mathrm{Cos}\left[\frac{\sqrt{k}\ t}{\sqrt{m}}\right] + \frac{\sqrt{m}\ v0\ \mathrm{Sin}\left[\frac{\sqrt{k}\ t}{\sqrt{m}}\right]}{\sqrt{k}}\right\}\right\}$

V zapisu je prvi odvod funkcije po argumentu označen z ', drug odvod funkcije po argumentu označen z '', presledek med posameznimi členi pa pomeni njihovo množenje.

V rezultatu nastopa izraz $\frac{\sqrt{k}}{\sqrt{m}}$, ki predstavlja lastno frekvenco ω. Da se doseže zapis, ki je inženirsko bolj uporaben, se v rezultatu izraz $\frac{\sqrt{k}}{\sqrt{m}} = \sqrt{\frac{k}{m}}$ zamenja s simbolom ω, rezultat pa se shrani pod spremenljivko npr. *rešitev*:

```
In[2]:= resitev = Expand[%] /. √k / √m → ω
```

Out[2]= $\left\{\left\{u[t] \to u0\ \mathrm{Cos}[t\,\omega] + \frac{\sqrt{m}\ v0\ \mathrm{Sin}[t\,\omega]}{\sqrt{k}}\right\}\right\}$

Ukaz *izraz1/.izraz2→ izraz3* pomeni, da se zgolj v tej ukazni vrstici v spremenljivki *izraz1 izraz2* nadomesti z *izrazom3*, izven te vrstice pa ostaja *izraz2* nespremenjen. Puščico je mogoče dobiti tudi z zapisom ->, ki ga Mathematica nato sama pretvori v →.

Rezultat je že bližje znanemu inženirskemu rezultatu, vendar *Mathematica* ni zaznala, da velja hkrati tudi $\frac{\sqrt{m}}{\sqrt{k}} = \frac{1}{\omega}$, kar očitno zahteva nov korak:

```
In[3]:= resitev = resitev /. √m / √k → 1/ω
```

Out[3]= $\left\{\left\{u[t] \to u0\ \mathrm{Cos}[t\,\omega] + \frac{v0\ \mathrm{Sin}[t\,\omega]}{\omega}\right\}\right\}$

Sedaj se končno lahko definira funkcija nedušenega pomika *un(t)* kot funkcija časa in sicer tako, da se kot funkcija definira izraz, podana v spremenljivki *resitev*:

```
In[4]:= un[t_] = u[t] /. resitev[[1]]
```

Out[4]= $u0\,\mathrm{Cos}[t\,\omega] + \frac{v0\,\mathrm{Sin}[t\,\omega]}{\omega}$

Čeprav ni nujno potrebno, se je mogoče prepričati, da funkcija *un(t)* zadošča obema začetnima oz. robnima pogojema:

kontrole

```
In[5]:= un[0]
```

Out[5]= u0

```
In[6]:= un'[0]
```

Out[6]= v0

kjer se je vrstica *kontrole* dodala zgolj kot komentar (text) z ukazom *Format/Style/Text* ali z *Alt+7*. Preveriti pa je tudi mogoče, ali dobljena rešitev diferencialne enačbe resnično ustreza diferencialni enačbi:

```
In[7]:= m un''[t] + k un[t]
```

Out[7]= $k\left(u0\,\mathrm{Cos}[t\,\omega] + \frac{v0\,\mathrm{Sin}[t\,\omega]}{\omega}\right) + m\,(-u0\,\omega^2\,\mathrm{Cos}[t\,\omega] - v0\,\omega\,\mathrm{Sin}[t\,\omega])$

Dobljeni rezultat na prvi pogled ne ustreza začetni homogeni diferencialni enačbi, vendar je potrebno upoštevati, da je *Mathematica* zgolj zapisala posamezne člene rezultata brez poskusa njegove ureditve, kar je potrebno zahtevati posebej, npr. z ukazom *Simplify[izraz]*:

```
In[8]:= Simplify[%]
```

Out[8]= $\frac{(k - m\,\omega^2)\,(u0\,\omega\,\mathrm{Cos}[t\,\omega] + v0\,\mathrm{Sin}[t\,\omega])}{\omega}$

Rezultat je sicer že bistveno krajši, vendar je potrebno upoštevati še zvezo $\omega^2 = \frac{k}{m}$:

```
In[9]:= Simplify[% /. ω^2 → k / m]
```

Out[9]= 0

kar kočno potrdi korektnost rešitve.

3. Reševanje diferencialne enačbe lastnega dušenega nihanja

Diferencialna enačba lastnega dušenega nihanja pa se zapiše kot:

$$m \cdot \ddot{u} + c \cdot \dot{u} + k \cdot u = 0$$

kjer je *c* sedaj koeficient dušenja. Še vedno gre za diferencialno enačbo drugega reda, kar pomeni, da sta tudi sedaj potrebni dve informaciji (robna pogoja), ki omogočata določitev neznanih integracijskih konstant. Če se ponovno uporabita najbolj pogosta splošna začetna pogoja: začetni (t=0) pomik u(0) $\equiv u_0$ in začetno hitrost $\dot{u}(0) = v_o$, se v *Mathematici* diferencialna enačba, skupaj z začetnima pogojema, sedaj zapiše kot (še prej pa se definira nova sekcija z nazivom *Dušeni odziv*):

■ Dušeni odziv

```
In[10]:= resitevdusena =
    DSolve[{m u''[t] + c u'[t] + k u[t] == 0, u[0] == u0, u'[0] == v0}, u[t], t]
```

Out[10]=
$$\left\{\left\{u[t] \to \frac{e^{\frac{\left(-c-\sqrt{c^2-4km}\right)t}{2m}}\left(-c\,u0+\sqrt{c^2-4km}\,u0-2\,m\,v0\right)}{2\sqrt{c^2-4km}} + \frac{e^{\frac{\left(-c+\sqrt{c^2-4km}\right)t}{2m}}\left(c\,u0+\sqrt{c^2-4km}\,u0+2\,m\,v0\right)}{2\sqrt{c^2-4km}}\right\}\right\}$$

Rešitev je sedaj nekoliko daljša in posledično manj pregledna, pa tudi uporaba ukaza *Simplify[]* ne prinese napredka:

```
In[11]:= Simplify[%]
```

Out[11]=
$$\left\{\left\{u[t] \to \frac{1}{2\sqrt{c^2-4km}}\left(e^{-\frac{\left(c+\sqrt{c^2-4km}\right)t}{2m}}\left(c\left(-1+e^{\frac{\sqrt{c^2-4km}\,t}{m}}\right)u0 + \left(1+e^{\frac{\sqrt{c^2-4km}\,t}{m}}\right)\sqrt{c^2-4km}\,u0 + 2\left(-1+e^{\frac{\sqrt{c^2-4km}\,t}{m}}\right)m\,v0\right)\right)\right\}\right\}$$

kot tudi ne poskus pretvorbe eksponentnih funkcij v trigonometrične z ukazom *ExpToTrig[izraz]*:

```
In[12]:= ExpToTrig[%]
```

Out[12]=
$$\left\{\left\{u[t] \to \frac{1}{2\sqrt{c^2-4km}}\left(e^{-\frac{\left(c+\sqrt{c^2-4km}\right)t}{2m}}\left(c\left(-1+e^{\frac{\sqrt{c^2-4km}\,t}{m}}\right)u0 + \left(1+e^{\frac{\sqrt{c^2-4km}\,t}{m}}\right)\sqrt{c^2-4km}\,u0 + 2\left(-1+e^{\frac{\sqrt{c^2-4km}\,t}{m}}\right)m\,v0\right)\right)\right\}\right\}$$

Ker pa je smisel te vaje (in tudi običajne inženirske analize) zgolj primerjava nedušenega in dušenega odziva (glej točko 4) in ne oblika zapisa sama, se funkcija ohrani kar v tej obliki (čeprav je iz literature znana elegantnejša oblika zapisa) in se definira funkcija dušenega nihanja npr. kot *ud(t)*:

```
In[13]:= ud[t_] = u[t] /. %[[1]];
```

4. Prikaz razlike med odzivoma nedušenega in dušenega lastnega nihanja

Prikaz odziva je mogoč le, če so podane numerične vrednosti parametrov. Uporabijo se npr. podatki iz vaje 7 (http://www.geocities.com/mcsDISK/e7.pdf), kjer se sicer obravnava vsiljeno nedušeno nihanje. Da se kljub odsotnosti obtežbe sedaj dobi nek odziv, se izbereta začetna vrednost pomika npr. $u_o=0.2$ m in začetna hitrost npr. $v_o = 0.1\frac{m}{s}$:

■ Zgled - podatki iz vaje 7

```
In[14]:= M1 = 2000;
         M2 = 1500;
         EI = 512 10^6;
         α = 45 / 180 π;
         L1 = 3 √2 ;
         L2 = 4;
```

S temi definiranimi podatki je mogoče že izračunati nekaj veličin, katerih analitična izpeljava je bila narejena v vaji (zaradi upoštevanja dušenja se vpelje koeficient ξ, ki predstavlja razmerje med dejanskim in kritičnim dušenjem $\xi = \frac{c}{c_{cr}}$, iz česar je mogoče dejansko dušenje izraziti s pomočjo kritičnega dušenja in koeficienta ξ kot $c = \xi \cdot c_{cr} = \xi \cdot \sqrt{k \cdot m}$:

```
In[20]:= Vb = L1 / (L1 Cos[α] + L2);
         k = 3 EI / L2^2 / Vb^2 / (L1 + L2);
         m = M1 + M2 (Sin[α])^2;
         c = ξ √(k m) ;
         u0 = 0.2;
         v0 = 0.1;
         ω = √(k/m)
```

$$\text{Out[26]= } 560\sqrt{\frac{10}{33\left(4+3\sqrt{2}\right)}}$$

Ker je lastna krožna frekvenca zapisana (sicer točno, a) v nepregledni obliki, se poišče numerična vrednost izraza (ki mora biti, kljub odsotnosti obtežbe, enaka kot v primeru 7) z ukazom *N[argument]*:

```
In[27]:= N[%]
Out[27]= 107.374
```

ter nato še frekvenca v Hz:

```
In[28]:= % / 2 / Pi
Out[28]= 17.0891
```

ter še perioda oz. nihajni čas:

```
In[29]:= T = 1 / %
Out[29]= 0.058517
```

ki služi zgolj kot orientacijska vrednost časa, v katerem se bo opazoval in primerjal odziv nedušenega in dušenega nihanja. Kot ta čas se npr. izbere petkratna vrednost periode T:

```
In[30]:= casodziva = 5 T;
```

Slika nedušenega odziva se izriše in imenuje npr. *neduseni*:

```
In[31]:= neduseni = Plot[un[t], {t, 0, casodziva}];
```



Za izris dušenega odziva pa je potrebno še definirati vrednost koeficienta ξ. Za tipične gradbene konstrukcije pa velja, da imajo sorazmerno majhno dušenje ($\xi < 0.2$, običajno pa $0.02 < \xi < 0.1$), za AB konstrukcije pa npr. velja $0.07 < \xi < 0.1$.

Zaradi tega bosta izrisani dve krivulji dušenega odziva. Prva za vrednost ξ=0.02, druga pa za drugo skrajnost, torej ξ=0.1. Čeprav se bo ta krivulja zaradi zmanjševanja amplitude skozi čas zaradi dušenja jasno razlikovala od odziva nedušenega nihanja, bo ta krivulja (predvsem iz demonstracijskih razlogov prikaza možnosti *Mathematice*) izrisana s črtkano črto, kar se doseže z opcijo npr. *PlotStyle -> {{Dashing[{0.1, 0.03}]}}*, ki sledi parametrom izrisa v ukazu *Plot[]*. Prva vrednost (0.1) podaja dolžino črtkanega dela krivulje, druga (0.03) pa dolžino praznega prostora med dvema črticama.

Vrednost parametra ξ se zgolj za izris te slike upošteva kot 0.02:

```
In[32]:= dusenimin = Plot[ud[t] /. ξ → 0.02, {t, 0, casodziva},
           PlotStyle → {{Dashing[{0.1, 0.03}]}}];
```



Druga praktično skrajna vrednost parametra ξ je 0.1, za izris pa je spet uporabljena drugačna vrsta krivulje (črta – pika – črta), kjer so podani dolžina črte (0.1), razmak med črto in piko (0.02), dolžina pike oz. nove črte (0.01) ter še razdalja med pika in novo črto (0.02):

```
In[33]:= dusenimax = Plot[ud[t] /. ξ → 0.1, {t, 0, casodziva},
           PlotStyle → {{Dashing[{0.1, 0.02, 0.01, 0.02}]}}];
```



Čeprav je iz posameznih slik že razvidna razlika med nedušenim in dušenim (celo z različnimi stopnjami) odzivom, se popolnoma jasen vpogled dobi šele z neposredno primerjavo vseh treh krivulj:

```
In[34]:= Show[neduseni, dusenimin, dusenimax];
```



Iz slike je jasno razvidno, da se amplituda pri dušenem odzivu postopoma zmanjšuje in da s časom limitira k nič, kar je še posebej jasno razvidno, če se slika odziva izriše na nekoliko daljšem časovnem intervalu:

■ Iskanje casa izdusenja

```
In[35]:= Plot[ud[t] /. ξ → 0.02, {t, 0, 5}, PlotRange → All];
```



še bolj drastično pa je to opazno pri večjem koeficientu dušenja:

```
In[36]:= Plot[ud[t] /. ξ → 0.1, {t, 0, 5}, PlotRange → All];
```



Zato se lahko postavi neka meja (npr. u(t)=0.1 % u_{max}), pri kateri se smatra, da se je odziv izdušil. Ta meja se izračuna kot ničla enačbe z ukazom *FindRoot[enačba,{spremenljivka, okvirna ničla}]*, ukaz *Chop[izraz]* pa zgolj poskrbi, da se ne izpišejo morebitne izredno majhne vrednosti, ki se pojavijo zaradi numerične analize.

```
In[37]:= Chop[FindRoot[(ud[t] /. ξ → 0.02) / u0 == 0.001, {t, 5}]]

Out[37]= {t → 4.96163}

In[38]:= Chop[FindRoot[(ud[t] /. ξ → 0.1) / u0 == 0.001, {t, 1}]]

Out[38]= {t → 1.00908}
```

Okvirna vrednost ničle funkcije se za vsako vrednost dušenja odčita iz pripadajoče slike.